

**Государственный комитет Российской Федерации
по высшему образованию**

Ульяновский государственный технический университет

ИМИТАЦИОННОЕ МОДЕЛИРОВАНИЕ НА ЯЗЫКЕ GPSS

методические указания

**Составители: О.Н.Евсеева
В.В.Шишкин**

Ульяновск 1995

УДК 681.03

Имитационное моделирование на языке GPSS : Метод. указ. /
О.Н.Евсеева, В.В.Шишкин. - Ульяновск: УлГТУ, 1995. - 40 с.

Настоящие методические указания разработаны в соответствии с программой курса "Моделирование" и предназначены для студентов специальности 2201, 5528.

Предлагаемый практикум может быть использован для обучения студентов методам построения дискретных имитационных моделей вычислительных систем различного типа средствами специального языка имитационного моделирования GPSS.

Рецензент _____

Рекомендован к изданию
методической комиссией РТФ

Ульяновский политехнический институт, 1995.

СОДЕРЖАНИЕ

ВВЕДЕНИЕ	4
1. ОСНОВНЫЕ СВЕДЕНИЯ О СИСТЕМЕ МОДЕЛИРОВАНИЯ	
1.1. Модели массового обслуживания	4
1.2. Назначение и состав системы моделирования	7
1.3. Сокращенное описание объектов GPSS	8
1.4. Протокол и результаты работы программы на GPSS	10
2. ПРИМЕНЕНИЕ GPSS ДЛЯ МОДЕЛИРОВАНИЯ ВЫЧИСЛИТЕЛЬНЫХ СИСТЕМ	12
2.1. Моделирование ВС, представленных как одноканальные системы массового обслуживания	13
2.2. Модель мультипрограммной ВС коллективного пользования	19
2.3. Модели ВС с изменяемой интенсивностью обслуживания задач и случайным выбором программы обработки	21
2.4. Модели ВС с различными дисциплинами обслуживания	24
2.5. Модель последовательной ВС	26
2.6. Модель параллельной ВС	27
3. ПРАКТИКУМ ПО МОДЕЛИРОВАНИЮ ВС НА GPSS	28
3.1. Общий порядок выполнения заданий	28
3.1.1. Самостоятельная подготовка студентов	28
3.1.2. Содержание отчета	28
3.2. Задания практикума	29
3.2.1. Моделирование случайных величин	29
3.2.2. Моделирование ВС как СМО с одним обслуживающим прибором	34
3.2.3. Моделирование многопроцессорных ВС с перераспределением потока заявок	35
3.2.4. Формирование потоков заявок ВС	37
ЛИТЕРАТУРА	40

В В Е Д Е Н И Е

Имитационное моделирование - это процесс конструирования модели реальной системы и постановки экспериментов на этой модели с целью либо понять поведение системы, либо оценить различные стратегии управления, обеспечивающие функционирование системы в заданных границах [6].

Реализация имитационных моделей на ЭВМ производится в настоящее время, как правило, с помощью общецелевых или специализированных языков моделирования. К общецелевым языкам моделирования относится язык GPSS, ориентированный на исследование систем массового обслуживания (СМО). GPSS является самым распространенным в мире языком моделирования (уступает лишь СИМУЛе) [7].

Математические модели и методы анализа ВС строятся на основе СМО. Анализ заключается в определении выходных параметров ВС при заданных потоках задач, поступающих для решения в систему. Аналитические модели удается получить в сравнительно простых частных случаях, которые характеризуются рядом упрощающих предположений.

Имитационная модель позволяет воспроизвести обслуживание в системе множества задач при любых заданных статистических сведениях о входных потоках и ресурсах ВС.

1. ОСНОВНЫЕ СВЕДЕНИЯ О СИСТЕМЕ МОДЕЛИРОВАНИЯ

1.1. Модели массового обслуживания

В моделях массового обслуживания рассматриваются системы, на вход которых в произвольные моменты времени поступают заявки, требующие обслуживания путем предоставления определенных ресурсов на некоторое время и покидающие систему после обслуживания.

Системы массового обслуживания делятся на одноканальные - с одним обслуживающим прибором (ресурсом) и многоканальные - содержащие несколько обслуживающих приборов (рис 1.1). Для создания СМО необходимо задать:

- распределение длительности интервалов времени между заявками входного потока;





Рис. 1.1. Системы массового обслуживания:
а) одноканальная; б) многоканальная

- число обслуживающих приборов;
- распределение длительности обслуживания заявок обслуживающими приборами;
- стратегию выбора заявок из очереди.

Для заданного множества параметров СМО можно получить такие характеристики системы, как: время обслуживания заявки в системе; время ожидания заявкой обслуживания (время нахождения в очереди); длина очереди; загрузка обслуживающих приборов; длительность простоя обслуживающих приборов и т.д.

Очевидно, что поскольку как минимум один из параметров модели имеет вероятностный характер, то и характеристики СМО также будут вероятностными. То есть могут быть представлены как минимум на двух уровнях: статистического ряда и среднего значения. Обычно для характеристики типа СМО используют обозначение вида $X/Y/k$, где первый параметр обозначает закон распределения времени между поступлением заявок, второй параметр - закон распределения длительности обслуживания заявок и третий параметр - число обслуживающих приборов. При задании законов распределения традиционно M обозначает экспоненциальный, а G - произвольный закон распределения. Кроме экспоненциального и произвольного законов распределения могут применяться эрланговский, нормальный и другие законы.

В теории массового обслуживания СМО могут объединяться путем передачи заявок с выходов одних на входы других СМО, образуя таким образом сети массового обслуживания. Для задания сети массового обслуживания требуется определить все входящие в нее СМО, а также определить конфигурацию связей систем и вероятности передачи заявок между ними. Естественно, что сумма вероятностей передачи заявок из любой СМО должна равняться 1.

Обычно сеть массового обслуживания представляют в форме графа, где каждая из вершин графа является СМО (рис. 1.2). При анализе системы с помощью сети массового обслуживания можно оценивать каждую СМО по приведенным выше характеристикам, а также можно оценивать характеристики сети в целом. В качестве характеристик сети можно применять: загрузку сети; время обслуживания заявки в сети; время ожидания заявкой обслуживания (время нахождения в очередях); число заявок, находящихся в очередях, в обслуживающих приборах и в сети в целом; длительность простоя обслуживающих приборов и т.д.




Рис. 1.2. Сеть массового обслуживания

Как видно из описания, модели массового обслуживания хорошо подходят для имитации функционирования вычислительных систем, локальных и глобальных вычислительных сетей.

В самом деле, на вход любой вычислительной системы поступают входные потоки задач, которые требуют для своего решения определенных ресурсов. Моменты поступления задач в общем случае не детерминированы. Случайными также являются и требуемые задачами ресурсы: объем памяти, время процессоров, устройства ввода/вывода и т.д. В системе могут возникать отказы и сбои затребованных ресурсов, что также является случайными событиями.

Проектировщик, меняя конфигурацию системы: количество и типы процессоров, банков памяти, интерфейсов, организацию системы, принципы и стратегии обработки и т.д., может получать с помощью разработанной модели массового обслуживания временные характеристики проектируемой системы и характеристики для оценки ее надежности. Полученные характеристики являются для проектировщика основой для выбора наиболее эффективной конфигурации системы, принципов и стратегий обработки.

1.2. Назначение и состав системы моделирования

Общецелевая система моделирования GPSS (GENERAL PURPOSE SIMULATING SYSTEM) предназначена для построения имитационных моделей дискретных сложных систем различной физической природы.

В состав системы GPSS входят: язык моделирования GPSS; интерпретатор.

Имитационная модель на языке GPSS представляет собой список объектов (операторов или блоков), характеризующих процессы обработки заявок - пути их продвижения в моделируемой системе и сопровождающие это продвижение события. С помощью объектов языка описываются возникновение транзактов (заявок), их продвижение, задержки в очередях и устройствах, уничтожение и т.д. Последовательность записи объектов и специальные операторы управления задают маршруты перемещения транзактов между элементами модели.

Реализация работы отдельных объектов, их синхронизация и собственно моделирование осуществляется автоматически при помощи комплекса служебных программ GPSS. В процессе моделирования организуется сбор статистических данных, которые позволяют определить загрузку устройств, заполненность очередей, средние затраты времени на обработку, ожидание и т.п.

Объекты языка GPSS могут быть разделены на четыре класса: (1) динамические, (2) статические или оборудование, (3) статистические, (4) операционные.

Динамические объекты (транзакты) в процессе моделирования могут возникать и уничтожаться. С каждым транзактом связан ряд параметров, которым пользователь может присваивать различные значения, характеризующий транзакт.

Статические объекты служат для описания аппаратных средств системы, моделируемой на GPSS. К ним относятся: одноканальные устройства, многоканальные устройства (накопители или память).

Одноканальное устройство может обслуживать только один транзакт. Накопитель может обслуживать одновременно несколько транзактов. Статистические объекты (очереди и таблицы) служат для оценки поведения моделируемой системы. В очереди хранится перечень транзактов, задержанных в каком-либо месте исследуемой системы и длительности этих задержек. Таблицы, в основном, предназначены для использования при построении статистических распределений выбранных для анализа величин.

Операционные объекты задают логику работы модели и определяют пути следования и действия транзактов.

Помимо перечисленных объектов GPSS включает ряд дополнительных элементов, таких как описания накопителей и таблиц, команды управления, сохраняемые величины, матрицы, арифметические переменные, функции и т.д.

Развитие любой реальной системы происходит во времени. В от-

личии от реальных систем, в GPSS время изменяется дискретно. Соотношение единицы времени моделируемой системы и единицы времени в программе моделирования определяется пользователем и должно быть целым числом, например, единица времени модели может соответствовать микросекунде, минуте, часу и т.д. В GPSS-моделях введены понятия абсолютного и относительного времени. Абсолютное время начинает свой отсчет с момента начала моделирования, а относительное - с момента определяемого пользователями как момент, начиная с которого необходимо собирать статистику о ходе моделирования. Если нет указаний пользователя, абсолютное и относительное время в GPSS совпадают.

В GPSS-моделях транзакт в любой момент времени может находиться в одном из пяти списков.

В "списке текущих событий" находятся те транзакты, время наступления которых меньше (транзакты, которые должны были начать двигаться в некоторый момент в прошлом, но были заблокированы) или равно текущему времени.

"Список будущих событий" содержит транзакты, время поступления которых больше текущего времени, то есть события, которые должны произойти в будущем.

"Список прерываний" содержит те отдельные транзакты, обслуживание которых соответствующими устройствами было прервано, а также те которые вызвали прерывание.

"Список синхронизируемых сообщений" содержит транзакты, находящиеся в данный момент в состоянии сравнения.

"Список пользователя" содержит транзакты, удаленные пользователем из списка текущих событий и помещенные в список пользователя как временно неактивные.

Списки текущих и будущих событий являются основными. Каждый транзакт продвигается в модели до тех пор, пока его путь не будет заблокирован занятым блоком или пока транзакт не войдет в блок, задающий временную задержку. Из задержанных транзактов образуется список будущих событий, а из заблокированных транзактов - список текущих событий. Просматривая список текущих событий, система моделирования пытается передвинуть содержащиеся в нем транзакты к следующим блокам модели. Если продвижение транзактов не возможно, то моделируемое системное время сдвигается к моменту наиболее раннего из будущих событий. При этом соответствующий транзакт переносится в список текущих событий и передвигается, если имеется

возможность, дальше по схеме модели. В том случае, если рассматриваемый транзакт входит в блок задержки, то он снова попадает в список будущих событий. В случае блокирования транзакт остается в списке текущих событий. Работа моделирующей системы, в основном сводится к периодическому повторению описанных выше действий и прекращается при окончании процесса моделирования.

1.3. Сокращенное описание объектов GPSS

В подполях операторов языка могут записываться числа, стандартные числовые атрибуты (СЧА), последовательности букв и цифр, обозначающие метки и идентификаторы, вводимые пользователем. СЧА отображают свойства и состояния объектов моделируемой системы (оборудования, очередей и т.д.). Каждый СЧА имеет определенное мнемоническое обозначение, задающие тип атрибута, за которым иногда следует числовое или символьное имя объекта. Числовое имя содержит до пяти цифр, символьное - от трех до пяти символов, из которых первые три должны быть буквами.

Ниже описание объектов языка и СЧА дано в виде примеров с конкретными числовыми значениями и обозначениями меток и идентификаторов. Более полное описание и применение объектов и СЧА GPSS приведено в разделе 2.

Примеры записи СЧА :

V15 - переменная с числовым именем 15.

200 - целочисленная константа, значение которой равно 200.

FN5 - вычисленное значение функции с числовым именем 5.

FN#RES - вычисленное значение функции с символьным именем RES.

RN2 - вычисляемое с помощью второго датчика псевдослучайное число, равномерно распределенное в интервале от 0 до 999 включительно, если только эта величина не будет использована как переменная или как функция. В этом случае эта величина принимает любое значение в диапазоне 0 - 0,999999.

Примеры записи основных операторов языка GPSS, описаний и управления. GENERATE 100 - генерация транзактов с постоянным интервалом следования, равным 100 единиц модельного времени.

QUEUE COR - организация очереди COR, ее длина увеличивается на единицу.

DEPART CUR - длина очереди CUR уменьшается на 1.

FNC FUNCTION RN3, C4 0,0/.4,1.2/1,2/2,4

- описание функции FNC, ее аргументом является случайная величина (датчик RN3), равномерно распределенная в диапазоне [0,0.999999], функция непрерывная (C) и задана таблично четырьмя узловыми точками (0,0),(0.4,1.2),(1,2),(2,4).

FE FUNCTION P2, D3 .2,40/.6,100/1,150

- описание дискретной числовой функции с именем FE, заданной таблично в трех узловых точках, аргументом которой является текущее значение второго параметра транзакта.

END - последняя строка в программе на GPSS, обеспечивающая прекращение моделирования.

1.4. Протокол и результаты работы программы на GPSS

Программа производит вывод информации на печать в трех случаях: после нормального завершения моделирования, когда значение счетчика завершений, первоначально задаваемого командой START, стало равным нулю или меньше нуля; после обнаружения ошибки.

Выдаваемая информация представляет собой печать статистики об устройствах, накопителях, очередях.

Статистика об устройствах выдается для всех устройств, к которым производилось обращение в процессе моделирования. Эта статистика содержит коэффициенты использования (часть времени, в течении которого устройство было занято или обслуживало прерывания), общее число обращений к устройству, среднее время, в течении которого устройство занято одним транзактом, и номер транзакта, обслуживаемого устройством в момент окончания счета. Этот транзакт мог либо захватить устройство в момент, когда оно свободно, либо прервать обслуживание на этом устройстве другого транзакта. При распечатке указывается каким из способов транзакт захватил устройство.

Статистика о накопителях выдается для всех накопителей, обращение к которым производилось в процессе моделирования. Статистика для каждого накопителя содержит: емкость накопителя, заданную пользователем; среднее содержимое накопителя за время моделирования (среднее число занятых единиц памяти накопителя за время моделирования); среднее использование накопителя за время модели-

рования (среднее содержимое, отнесенное к максимальной емкости);общее число входов в память (накопитель); среднее время пребывания транзакта в накопителе; текущее содержимое накопителя в момент окончания счета; максимальное содержимое накопителя, зафиксированное за время моделирования.

Статистика очередей представляет информацию обо всех очередях, обращении к которым производилось в процессе моделирования.

Для каждой очереди статистика содержит: максимальную длину очереди за время моделирования; среднюю длину очереди; общее число входов в очередь; число транзактов, которые вошли в очередь, но не были в ней задержаны (число "нулевых входов"); среднее время пребывания транзакта в очереди; среднее время пребывания в очереди задержанных транзактов (т.е. подсчитанное без учета нулевых входов); номер таблицы, соответствующей очереди, если такая таблица была задана описанием QTABLE; текущая длина очереди в момент окончания счета.

2. ПРИМЕНЕНИЕ GPSS ДЛЯ МОДЕЛИРОВАНИЯ ВЫЧИСЛИТЕЛЬНЫХ СИСТЕМ

При моделирование на GPSS ВС представляется как система массового обслуживания. Элементы СМО делятся на два основных класса: объекты обслуживания - в ВС это решаемые задачи, называемые в GPSS транзактами; и средства, называемые в GPSS статическими объектами.

Имитационное моделирование используется как при анализе, так и при синтезе ВС. Здесь рассматривается только задача анализа. Задача анализа формулируется как задача определения выходных параметров ВС выбранной структуры при заданных параметрах аппаратуры выходные параметры моделируемой задач. Средства позволяют определять выходные параметры моделируемой ВС (производительность,быстродействие,средние и максимальные времена и вероятности обслуживания,средние и максимальные длины очередей и т.п.) и вектор переменных состояния ее элементов. Ниже рассматриваются модели ВС,которые могут быть представлены различными типами СМО: одноканальными, многоканальными, многофазными, замкнутыми и т.д.

2.1. Моделирование ВС, представленных как одноканальные системы массового обслуживания

В этом разделе разрабатываются и исследуются GPSS-модели ВС, имеющей один вход и один выход. На вход ВС в случайные или детерминированные моменты времени поступают задания на обработку. Задания выполняются либо сразу, если ВС свободна, либо становятся в очередь. Время обработки задания также случайно или детерминировано. После обработки задания покидают систему. Схема прохождения потока заданий, обрабатываемых ВС, приведена на рис.1.1.а как типовая одноканальная СМО.

Пример 2.1

При разработке GPSS-модели этой системы необходимо осуществить программную имитацию следующих событий: создание задания (транзакта), постановку транзакта в очередь, уход транзакта из очереди, занятие ВС, задержку на время обработки (выполнения транзакта), освобождение ВС, уход транзакта из моделируемой системы. На рис.2.1 и 2.2 представлены два варианта GPSS-модели ВС с различным заданием длины интервала моделирования.

В приведенных на рис. 2.1 и 2.2 моделях задания (транзакты) поступают на обработку с интервалами, распределенными равномерно в диапазоне [15,35] (рис.2.1) или [5,25] (рис.2.2). Время обслуживания одного транзакта ВС распределено также равномерно в диапазоне [5,15]. Здесь и в дальнейшем используются безразмерные единицы модельного времени, которые могут соответствовать миллисекунде, секунде, минуте, и т.д. Модели включают в себя семь общих блоков, которые выполняют следующие функции:

- 1 - генерирование транзактов с равномерным распределением интервалов следования;
- 2 - постановка транзакта в очередь с числовым именем 1;
- 3 - занятие транзактом одноканального устройства с символическим именем SYST, имитирующим ВС с одним входом и одним выходом;
- 4 - выход транзакта из очереди;
- 5 - моделирование обслуживания транзакта устройства SYST;
- 6 - освобождение транзактом устройства SYST;
- 7 - удаление одного транзакта из модели.

Для модели, приведенной на рис.2.2, моделирование прекратится, когда блоком GENERATE будет сгенерировано количество транзак-

```
Номер  
блока операторы GPSS  
-----  
1 GENERATE 25,10  
2 QUEUE Q1  
3 SEIZE SYST  
4 DEPART Q1  
5 ADVANCE 10,5  
6 RELEASE SYST  
7 TERMINATE 1  
START 400  
END
```

Рис.2.1. Модель ВС с фиксированным количеством обслуженных транзактов.

```
Номер  
блока операторы GPSS  
-----  
1 GENERATE 15,10  
2 QUEUE Q1  
3 SEIZE SYST  
4 DEPART Q1  
5 ADVANCE 10,5  
6 RELEASE SYST  
7 TERMINATE  
8 GENERATE 4920  
9 TERMINATE 1  
10 START 1  
END
```

Рис.2.2. Модель ВС с фиксированным числом единиц модельного времени

тов, заданное в строке 9 START (в данном случае 400). В это количество входит число транзактов, находящихся в очереди и на обслуживании.

Модель рис.2.2 содержит помимо основного сегмента (блоки 1-7) дополнительный, который задает время моделирования (блоки 8-9, управляющая строка 10). Блоки GENERATE с номерами 1 и 8 работают независимо друг от друга. Через 4920 единиц модельного времени блок GENERATE с номером 8 сгенерирует один транзакт, что вызовет вычитание единицы из счетчика завершений (строки 9-10), и моделирование прекратится.

После прогона рассмотренных выше моделей на ЭВМ на печать выдается статистика, относящаяся к устройству SYST и очереди Q1 в виде (рис. 2.3, 2.4).

Для моделей (рис.2.1 и 2.2) она содержит следующие основные результаты соответственно:

- коэффициент загрузки ВС 0.402 и 0.660;
- число прошедших обработку транзактов 400 и 325;
- максимальная длина очереди 1 и 2;
- процент транзактов с нулевым временем ожидания в очереди 100 и 73.5;
- среднее время ожидания в очереди 0 и 1.474.

GPSS/PC. Общецелевая Система Моделирования. УлПИ. 1994 год. Моделирование завершено.

Приборы	Нагрузка	Число входов	Среднее время/тран.	Номер транзакта	Номер обл.
SYST	0.402366	400	10.117500	2	
Очереди	Макс. содерж.	Среднее содерж.	Общ. число входов	Число нул. вх.	Процент нул. вх.
Q1	1	1.000000	400	400	1.000000
Среднее время/тран.	\$Среднее время/тран.	Текущее содержимое			
0.000000	0.000000	0			

Рис. 2.3. Статистика для модели ВС (рис. 2.1).
GPSS/PC. Общецелевая Система Моделирования. УлПИ. 1994 год.

Моделирование завершено.

Приборы	Нагрузка	Число входов	Среднее время/тран.	Номер обсл. транзакта
SYST	0.660163	325	9.993846	3

Очереди	Макс. содерж.	Среднее содерж.	Общ. число входов	Число нул. вх.	Процент нул. вх.
Q1	2	1.000000	325	239	0.735385

Среднее время/тран.	\$Среднее время/тран.	Текущее содержимое
1.473846	5.569767	0

Рис. 2.4. Статистика для модели ВС (рис. 2.2).

Язык GPSS предоставляет пользователю средства, которые позволяют исследовать ВС при изменении ее параметров или параметров входного потока заданий. Для повышения точности моделирования целесообразно исследовать ВС в установившемся режиме, что также реализуется средствами GPSS.

Рассмотрим пример, иллюстрирующий эти возможности GPSS.

Пример 2.2

Пусть необходимо исследовать ВС при различном времени обработки заданий: 200 и 150 единиц. Входной поток заданий распределен по закону Эрланга первого порядка с параметром $\lambda = 0.0067$. В результате моделирования должны быть получены коэффициенты загрузки ВС и статистика о количестве и времени нахождения заданий в системе (в очереди и на обслуживании).

Модель этой ВС приведена на рис.2.5. В этой модели выборка случайных чисел, распределенных по закону Эрланга, формируется с помощью датчика равномерно распределенных чисел RN1 и вычисляемого объекта "функции" с именем ERLAN.

Строки 1-7 модели определяют функцию распределения Эрланга первого порядка. Моделирование наступления событий производится блоком GENEKATE, операнд А которого является величиной обратной параметру X.

Номер блока	о п е р а т о р ы	Номер G P S S	Номер строки
	ERLAN FUNCTION	RN1,C24	1
	0,0/.0176,.2/.0616,.4/.1219,.6/.1913,.8		2
	.2642,1/.3374,1.2/.4082,1.4/.4751,1.6/		3
	.5372,1.8/.5941,2/.6254,2.2/.6915,2.4		4
	.7326,2.6/.7689,2.8/.8008,3/.8641,3.5		5
	.9084,4/.9596,5/.9826,6/.9927,7		6
	.997,8/.9988,9/.9995,10		7
	TAB1 TABLE	Q1,0,1,20	8
	TAB2 TABLE	M1,0,500,20	9
1	GENERATE	150,FN\$ERLAN	10
2	QUEUE	ETM	11
3	SEIZE	PORT1	12
4	METKA ADVANCE	200	13
5	RELEASE	PORT1	14
6	TABULATE	TAB1	15
7	DEPART	ETM	16
8	TABULATE	TAB2	17
9	TERMINATE	1	18
	START	500,NP	19
	RESET		20
	START	1000	21
	CLEAR		22
4	METKA ADVANCE	150	23
	START	500,NP	24
	RESET		25
	START	1000	26
	END		27

Рис. 2.5 Модель ВС с изменяемыми параметрами, исследуемая в установившемся режиме

Строки 8-9 задают описание таблиц TAB1 и TAB2. С их помощью формируются гистограммы распределения числа транзактов и времени нахождения транзактов в исследуемой системе. Таблица с именем TAB1 ссылается на СЧА Q1-число транзактов между блоками QUEUE ETM

и DEPART ETM, то есть Q1-число транзактов в системе. СЧА M1 в таблице TAB2 означает время нахождения транзактов в модели. Блоки модели имеют следующее назначение:

- 1 - генерирование транзактов с интервалами следования, распределенными по закону Эрланга с параметром $X=0,0067$;
- 2 - вхождение в очередь с именем ETM;
- 3 - занятие транзактом устройства PORT1, имитирующего ВС;
- 4 - моделирование обслуживания задания ВС с временем обработки 200 единиц;
- 5 - освобождение транзактом устройства PORT1;
- 6 - формирование таблицы частот для СЧА Q1;
- 7 - выход транзакта из очереди ETM;
- 8 - формирование таблицы частот для СЧА M1;
- 9 - удаление транзакта из модели.

Команды в строках 19 и 20 обеспечивают сброс накопленных статистик после обслуживания 500 транзактов (работа в неустановившемся режиме) и запрет печати результатов (операнд В, заданный

Таблица 2.1
Параметры исследуемой ВС

п.п.	Определяемый параметр	Время T обработки одного задания ВС	
		150	200
1	Коэффициент загрузки	0.514	0.644
2	Максимальное число заданий, находящихся в системе (в очереди и на обслуживании)	4	6
3	Среднее значение числа заданий	1.409	1.767
4	Среднеквадратичное отклонение числа заданий	0.579	0.937
5	Среднее значение времени обработки одного задания (с учетом пребывания в очереди)	180.583	296.235
6	Среднеквадратичное отклонение времени обработки одного задания	55.187	150.937

в виде NP, в команде START). Команда в строке 21 обеспечивает распечатку сведений об устройстве PORT1 и очереди ETM, после прохождения 1000 транзактов через модель в установившемся режиме.

Таблица 2.2
Вероятность нахождения в моделируемой системе
1, 2, 3-х и т.д. заданий

Число заданий	Вероятность	
	T=150	T=200
1	0.634	0.482
2	0.325	0.347
3	0.038	0.112
4	0.003	0.040
5	0	0.018
6	0	0.001

Команда CLEAR (строка 23) устанавливает начальное состояние модели с изменением параметра блока ADVANCE с меткой МЕТКА (операнд A этого блока принимает значение, равное 150 единиц). После этого прогон модели повторяется и программа заканчивает свою работу. Результаты моделирования приведены в таблицах 2.1 и 2.2.

Данные, приведенные в таблицах 2.1 и 2.2. можно использовать для обоснованного выбора параметров входного потока заданий, вычислительной системы и объема буферного запоминающего устройства (ЗУ) для хранения заданий, находящихся в очереди на обслуживание.

2.2. Модель мультипрограммной ВС коллективного пользования

Мультипрограммные ВС коллективного пользования относятся к замкнутым СМО, в которых общее число запросов в течение всего интервала моделирования остается постоянным. Запросы циркулируют в СМО, последовательно изменяя свои состояния в моменты перехода от одной фазы обслуживания к другой. Например, ВС с несколькими терминалами, где пользователь, работающий за терминалом, может послать в ВС только один запрос до тех пор, пока ВС ему не ответит.

При построении GPSS-моделей подобных систем число транзактов в модели должно оставаться постоянным в течение всего интервала моделирования. Это может быть достигнуто с помощью блока GENERATE с операндом D, не равным нулю, и блока TRANSFER, который работает в безусловном режиме и возвращает транзакт в начало модели.

Пример 2.3

На рис.2.6 приведена GPSS-модель BC, в состав которой входит 16 терминалов. Время подготовки пользователем запроса распределено равномерно в интервале [500,1500]. Время обслуживания запроса вычислительной системой распределено по нормальному закону со средним 500 и среднеквадратичным отклонением 50.

В модели в строке 1 описан накопитель емкостью 16 единиц с именем TERM, который моделирует работу 16-ти терминалов. В строках 2-8 описана нормированная функция нормального распределения с именем NRAS. Значения нормально распределенной псевдослучайной величины со средним - 500 и среднеквадратичным отклонением - 50 в модели получают с помощью переменной NORM, которая описана в строке 9. Блоки модели имеют следующее назначение:

- 1 - генерирование 16-ти транзактов;
- 2 - вхождение одного транзакта в накопитель с именем TERM;
- 3 - моделирование подготовки запроса пользователем;
- 4 - освобождение транзактом накопителя TERM;
- 5 - вхождение транзакта в очередь на устройство COMP, имитирующее BC;
- 6 - занятие транзактом устройства COMP;
- 7 - выход транзакта из очереди;
- 8 - моделирование обработки запроса BC (в качестве операнда A блока ADVANCE используется арифметическая переменная с именем NORM);
- 9 - освобождение транзактом устройства COMP (окончание обработки запроса BC);
- 10 - переход транзакта на блок 2 с меткой ВАНТ.

Для организации моделирования в течение 100000 единиц модельного времени используется сегмент программы, состоящий из блоков

11, 12 и управляющей строки 22.

Номер блока	о п е р а т о р ы G P S S	Номер строки
	TERM STORAGE 16	1
	NRAS FUNCTION RN5,C25	2
	0,-5/.00003,4/.00135,-3/.00621,-2.5	3
	.02275,-2/.06681,-1.5/.11507,-1.2/.15899,-1	4
	.21186,-.8/-.7425,-.6/.34458,-.4/.42074,-.2	5
	.5,0/.57926,.2/.65542,.4/.72575,.6/.78814,.8	6
	.8413,1/.88493,1.2/.93313,1.5/.97725,2	7
	.9979,2.5/.99865,3/.99997,4/1,5	8
	NORM FVARIABLE 500 + 50 * FN\$HPAC	9
1	GENERATE ,,16	10
2	BAHT ENTER TERM,1	11
3	ADVANCE 1000,500	12
4	LEAVE TERM,1	13
5	QUEUE Q1	14
6	SEIZE COMP	15
7	DEPART Q1	16
8	ADVANCE V\$NORM	17
9	RELEASE COMP	18
10	TRANSFER ,BAHT	19
11	GENERATE 100000	20
12	TERMINATE 1	21
	START 1	22
	END	23

Рис. 2.6. Модель ВС коллективного пользования

2.3. Модели ВС с изменяемой интенсивностью обслуживания задач и случайным выбором программы обработки

Некоторые ВС могут быть представлены как управляемые СМО, то есть СМО, в которых интенсивность обслуживания изменяется в зависимости от числа запросов, имеющих в СМО.

Пример 2.4

Рассмотрим модель ВС, решающей задачи оперативного управления некоторым объектом. С объекта управления на ВС поступают информационные заявки, интенсивность обработки которых зависит от общего числа заявок, находящихся на обслуживании в ВС. При числе заявок больше десяти ВС производит укороченную обработку каждой заявки за время равное 100 единиц, в противном случае обработка выполняется за 600 единиц реального времени. Считается, что в этом режиме ВС обеспечивает безаварийную работу управляемого объекта. Информационные заявки поступают на вход ВС в случайные равномерно распределенные в интервале [200,600] моменты времени.

Модель ВС с изменяемой интенсивностью обслуживания представлена на рис. 2.7. Для имитации переменного времени обслуживания информационных посылок в модели используется блок ADVANCE, меняющий время задержки в зависимости от значения СЧА Q\$ETM1. Зависимость времени задержки от СЧА Q\$ETM1 задана дискретной функцией PRER (строки 1-2). При числе транзактов, находящихся между блоками QUEUE ETM1 и DEPART ETM1, не превышающем 10, блок ADVANCE осуществляет задержку транзакта на 600 единиц, иначе - на 100 единиц.

Номер блока	операторы	Номер G P S S строки
	PRER FUNCTION Q\$ETM1,D2	1
	10,600/200,100	2
1	GENERATE 400,200	3
2	QUEUE ETM1	4
3	QUEUE ETM2	5
4	SEIZE SYST	6
5	DEPARTE ETM2	7
6	ADVANCE FN\$PRER	8
7	RELEASE SYST	9
8	DEPART ETM1	10
9	TERMINATE 1	11
	START 1000	12
	END	13

Рис.2.7. Модель ВС с изменяемой интенсивностью обслуживания.

Назначение других блоков и команд модели было рассмотрено выше. После прогона модели на ЭВМ на печать будут выданы статистика об устройстве SYST, имитирующем ВС, и очередях ETM1 и ETM2.

Пример 2.5

Рассмотрим модель ВС со случайным выбором программ обработки информационных заявок. Пусть имеется 3 типа информационных заявок, каждая из которых обрабатывается по своей программе. Известно, что время обработки заявки 1-го типа составляет 100 единиц времени, 2-го - 200, и 3-го - 1000. Выбор программы обработки определяется типом заявки. Так как заранее неизвестно, какая именно заявка поступает на обработку, то осуществляется случайный выбор одной из программ. Задана вероятность поступления заявок различных типов: 1-го типа - 0.4, 2-го - 0.55, 3-го - 0.05. Требуется собрать статистику о системе при таком же потоке обрабатываемых посылок, как и в предыдущем примере. Модель для решения этой задачи представлена на рис.2.8.

Номер блока	операторы	Номер GPSS строки
	DISR FUNCTION RN5,D3	1
	0.4,100/0.95,200/1.,1000	2
1	GENERATE 400,200	3
2	QUEUE ETM1	4
3	QUEUE ETM2	5
4	SEIZE SYST	6
5	DEPARTE ETM2	7
6	ADVANCE FN\$DISR	8
7	RELEASE SYST	9
8	DEPART ETM1	10
9	TERMINATE 1	11
	START 1000	12
	END	13

Рис.2.8. Модель ВС со случайным выбором программ обработки

Для случайного выбора программ обработки в модели используется дискретная случайная функция DISR, принимающая значения 100

с вероятностью 0.4, значения 200 с вероятностью 0.55 и 1000 с вероятностью 0.05 (строки 1-2). Функция DISR использует значения случайных величин, вработываемых программным датчиком GPSS RN5.

BC имитируется устройством с именем SYST. Блок ADVANCE осуществляет задержку обрабатываемой заявки (транзакта) в зависимости от ее типа в соответствии со значением функции DISR.

2.4. Модели BC с различными дисциплинами обслуживания

Некоторые типы запросов, обрабатываемых BC, имеют преимущества перед запросами других типов. Например, стоимость задержки при обработке определенного вида требований может быть очень высокой, вследствие чего целесообразно иметь время задержки по возможности минимальным. В других случаях предпочтение при обработке, оказываемое запросам с малым временем обслуживания, может сократить длину очереди и, тем самым, уменьшить объем буферного запоминающего устройства для хранения запросов.

Пусть на обработку в BC поступает несколько потоков заявок. Заявки i -го потока называют запросами приоритета i . Заявки i -го потока имеют более высокий приоритет по сравнению с заявками j -го потока ($i < j$), если запросы i -го потока пользуются преимуществами перед запросами j -го потока. В соответствии с этими преимуществами, приоритеты могут быть разделены на относительные и абсолютные. Относительный приоритет имеет место, когда поступивший в BC запрос обрабатывается без прерывания другим запросом. Следующим обрабатываемым запросом является запрос с самым высоким приоритетом или, если таких запросов несколько, один из них, тот, который ждет обслуживания дольше остальных. Абсолютный приоритет характеризуется тем, что запрос с более высоким приоритетом прерывает обслуживание запроса с низким приоритетом.

Пример 2.6

Рассмотрим GPSS-модель BC, на вход которой поступают четыре пуассоновских потока заданий с относительными приоритетами и со следующими параметрами: $X_1=0.005$; $X_2=0.01$; $X_3=0.03$; $X_4=0.04$ заявок в ед.времени. Время обработки задания для этих четырех потоков подчиняется экспоненциальному распределению с параметром $\mu=0.1$ заявок в ед.времени. Эта модель приведена рис.2.9.

Номер блока	операторы GPSS	Номер строки
	SYST EQU 1,F	1
	EXPON FUNCTION RN2,C24	2
	0,0/.1,.104/.2,.222/.3,.355/.4,.509/.5,.69	3
	.6,.915/.7,1.2/.75,1.38/.8,1.6/.84,1.83	4
	.88,2.12/.9,2.3/.92,2.52/.94,2.81/.95,2.99	5
	.96,3.2/.97,3.5/.98,3.9/.99,4.6/.95,5.3	6
	.998,6.2/.999,7/.9997,8	7
1	GENERATE 200, FN\$EXPON,,,4	8
2	TRANSFER ,MET1	9
3	GENERATE 100, FN\$EXPON,,,3	10
4	TRANSFER ,MET1	11
5	GENERATE 50, FN\$EXPON,,,2	12
6	TRANSFER ,MET1	13
7	GENERATE 25, FN\$EXPON,,,1	14
8	MET1 ASSIGN 1, SYST	15
9	QUEUE P1	16
10	SEIZE SYST	17
11	DEPART P1	18
12	ADVANCE 10, FN\$EXPON	19
13	RELEASE	20
14	TERMINATE 1	21
	START 1000	22
	END	23

Рис.2.9 Модель ВС с приоритетным обслуживанием

В строке 1 устройству SYST назначен номера 1. Нормированное экспоненциальное распределение описано строками 2-7. Блоки с номерами 1, 3, 5 и 7 генерируют транзакты с приоритетами, равными соответственно 1, 2, 3, 4, образующими пуассоновские потоки с заданными параметрами. Значения приоритетов в GPSS располагаются в порядке, обратном принятому в теории массового обслуживания ($1 < 2 < 3 < 4$). Блок 8 выполняет присваивание параметру 1 транзакта значения, соответствующего номеру устройства SYST, следовательно, очереди будет присвоен номер 1 (см.блоки 9 и 11). Назначение других блоков модели было рассмотрено выше.

2.5. Модель последовательной ВС

Рассмотрим модель ВС, состоящей из 1-го процессора предварительной обработки информационных заявок и 2-го процессора, входным потоком заданий для которого служит выходной поток первого.

Такая ВС осуществляет последовательную обработку информации и относится к многофазовым СМО.

Пример 2.7

Номер блока	операторы GPSS	Номер строки
	COMP1 EQU 1,F	1
	COMP2 EQU 2,F	2
	... строки определения ...	
	... функции EXPON см.рис.2.9 ...	
	TABT TABLE M1,0,100,1000	9
1	GENERATE 100,FN\$EXPON	10
2	ASSIGN 1,COMP1	11
3	QUEUE P1	12
4	SEIZE COMP1	13
5	DEPART P1	14
6	ADVANCE 50,FN\$EXPON	15
7	RELEASE COMP1	16
8	ASSIGN 2,COMP2	17
9	QUEUE P2	18
10	SEIZE COMP2	19
11	DEPART P2	20
12	ADVANCE 50,40	21
13	RELEASE COMP2	22
14	TABULATE TABT	23
15	TERMINATE 1	24
	START 1000	25
	END	26

Рис.2.10 Модель последовательной ВС

На рис.2.10. представлена GPSS-модель этой системы. Входной поток заданий является пуассоновским с параметром $\lambda=0,01$ заявок в

ед.времени. Время обработки задания для 1-го процессора распределено по экспоненциальному закону, для 2-го - по равномерному в интервале [10,90]. Результатом прогона модели являются статистика относящиеся к обоим процессорам и очередям к ним и таблица частот для времени обработки задания ВС.

2.6. Модель параллельной ВС

Моделирование параллельных ВС рассмотрим на примере ВС, состоящей из двух микропроцессорных систем (МПС), включающих в себя четыре однородных процессора.

Пример 2.8

Номер блока	операторы GPSS	Номер строки
...	строки определения функции EXPON	см.рис.2.9 ...
	STORAGE S\$MPC1,4/S\$MPC2,4	7
1	GENERATE 5,FN\$EXPON	8
2	TRANSFER .7,MPC2,MPC1	9
3	MPC1 QUEUE Q1	10
4	ENTER MPC1	11
5	ADVANCE 3	12
6	LEAVE MPC1	13
7	DEPART Q1	14
8	TRANSFER ,KOH	15
9	MPC2 QUEUE Q2	16
10	ENTER MPC2	17
11	ADVANCE 3	18
12	LEAVE MPC2	19
13	DEPART Q2	20
14	KOH TERMINATE 1	21
	START 500,NP	22
	RESET	23
	START 1000	24
	END	25

Рис.2.11. Модель параллельной ВС

Поток заданий на обработку является пуассоновским со значением среднего интервала 5 ед.времени. МПС осуществляют многоканальное обслуживание заданий. Время обработки задания процессором МПС равно 3 ед.времени. В среднем 70% заданий обрабатывается первой МПС, остальные - второй. Требуется в результате моделирования собрать статистику о МПС и об очередях к ним. GPSS-модель этой ВС представлена на рис.2.11.

После прогона этой модели на ЭВМ получены результаты, позволяющие сделать следующие выводы. Коэффициент использования процессоров первой МПС равен 0.113, второй - 0.045. Максимальное число заданий, обрабатываемых первой МПС, равно 5 (т.е. в очереди находилось не более одного задания). Максимальное число параллельно работающих процессоров второй МПС равно 3, т.е. в процессе работы ВС один из 4-х процессоров второй МПС не использовался.

3. ПРАКТИКУМ ПО МОДЕЛИРОВАНИЮ ВС НА GPSS

3.1. Общий порядок выполнения заданий

3.1.1. Самостоятельная подготовка студентов

Подготовка заключается в следующем:

- 1). Изучить теоретический материал, необходимый для выполнения задания.
- 2). Разработать модель исследуемой системы.
- 3). Составить GPSS программы моделирования.
- 4). Составить отчет и подготовиться к сдаче предыдущего задания.

3.1.2. Содержание отчета

Отчет должен содержать:

- 1). Постановку задачи моделирования.
- 2). Содержательное описание модели и исходные данные.
- 3). Таблицу определений GPSS-модели.
- 4). Выводы по результатам моделирования.

3.2. Задания практикума

3.2.1. Моделирование случайных величин

При подготовке к выполнению задания необходимо изучить методы получения случайных величин с заданными законами распределения

[4-8] и описания объектов языка GPSS [6,8]:

операторов - GENERATE A,B,C,D,E,F,G (ГЕНЕРИРОВАТЬ),
TERMINATE A (ЗАВЕРШИТЬ),
TABULATE имя (ТАБУЛИРОВАТЬ),
команд - START A,B,C,D (НАЧАТЬ),
END (ЗАКОНЧИТЬ),
определений - FUNCTION A,B (определение
X1,Y1/X2,Y2/./Xn,Yn. функции),
VARIABLE <арифм.выраж.> (переменная) или
FVARIABLE <арифм.выраж.>.
имя TABLE A,B,C,D (таблица)

Необходимо учитывать, что в GPSS имеются восемь датчиков равномерно распределенных псевдослучайных чисел, у каждого из которых есть свое имя: RN1, RN2,..., RN8. Имена датчиков являются стандартными числовыми атрибутами (СЧА), которые используются для получения псевдослучайных чисел с другими распределениями. При i -м обращении к любому из датчиков вычисляется произведение $a(k)*X(i-1)$, где k - индекс, который с равной вероятностью принимает значения 0,1,2,...,7; $a(k)$ - ядро; $X(i-1)$ - значение множителя, которое получено при предыдущем обращении к датчику. Если $i=1$, то $X(i-1)=X(0)$ - начальное значение множителя $X(i)$.

Дробные числа датчик выдает, если его имя используется в качестве аргумента функции. При использовании датчика в любом другом контексте выдаются целые числа, распределенные равномерно в интервале [0,999]. У всех восьми датчиков начальное значение индекса $k=0$, а начальное значение множителя $X(0)=1$. Поэтому все датчики формируют совершенно одинаковые псевдослучайные последовательности. Однако пользователь может изменить последовательность псевдослучайных чисел путем изменения начального значения $X(0)$ множителя. Такую возможность обеспечивает команда RMULT (УСТАНОВИТЬ ЗНАЧЕНИЯ ДАТЧИКОВ), имеющая восемь операндов от A до H,

которые используются для записи начальных значений множителя соответственно 1-8 датчиков. Так, например,

RMULT 325,14159,,8677

устанавливает начальные значения множителей датчиков RN1, RN2 и RN4 соответственно равными 325,14159,8677. Начальные значения множителей, задаваемых в RMULT, должны быть нечетными и иметь от одного до пяти разрядов.

Следует учитывать, что для имитации дискретных случайных величин используются дискретные функции GPSS (Пример 2.5).

Необходимо знать, что непрерывные случайные величины имитируются в GPSS моделях с помощью непрерывных функций. При этом следует использовать кусочно-линейную аппроксимацию функций, обратных нормированным функциям распределений. Определения функций для моделирования случайных величин, распределенных по экспоненциальному закону (EXPON) и нормальному (NORM) закону, приведены в Примерах 2.3, 2.6-8.

Содержание задания

Необходимо осуществить в соответствии с вариантом задания:

1. Моделирование дискретных случайных величин по заданному закону распределения;
2. Моделирование непрерывных случайных величин с равномерным, нормальным, экспоненциальным и пуассоновским распределениями.
3. Моделирование непрерывных случайных величин с заданным линейным законом распределения.
4. Проверку соответствия теоретических и полученных в результате моделирования распределений случайных величин.

Пример 3.1

Вариант задания описывается таблицей 3.1. В таблице 3.1 приняты следующие обозначения: y_1, y_2, \dots, y_n и p_1, p_2, \dots, p_n - значения дискретной случайной величины и соответствующие этим значениям вероятности; m - математическое ожидание; r - половина размаха равномерно распределенной случайной величины; λ - параметр экспоненциального распределения; σ - среднее квадратичное отклонение.

Таблица 3.1
Распределение

ряд	равномерное	нормальное	экспоненциальное
$Y_1 Y_2 \dots Y_n$	$m \ r$	$m \ \sigma$	λ
$P_1 P_2 \dots P_n$			
100 140 200	90 40	30 4	0.015
0.6 0.15 0.25			

Программа

Функции и переменные, определенные в программе, имеют следующие назначения: EXPON - для имитации значений экспоненциально распределенной случайной величины, имеющей интенсивность $\lambda=1$;

DSB - для имитации дискретной случайной величины, ряд распределения которой задан в таблице 3.1;

NORM - для имитации значений нормированной нормально распределенной случайной величины;

RAWN - переменная, обеспечивающая преобразование значений базовой псевдослучайной величины в значения равномерно распределенной псевдослучайной последовательности с $m=90$ и $r=40$;

GAUSS - переменная, которая обеспечивает преобразование значений нормированной нормальной псевдослучайной величины в значения нормально распределенной величины с $m=30$ и $\sigma=4$.

С помощью таблиц TAB1-TAB4 формируются векторы эмпирических частот и частотностей попадания псевдослучайных величин в каждый из D интервалов. В целях оценки адекватности непрерывного распределения псевдослучайной величины обычно задают $D= 10-20$. Значения операндов B,C,D таблицы TAB2 выбираются так, чтобы каждому интервалу принадлежало одно из возможных значений дискретной случайной величины (см. табл. 3.2).

* МОДЕЛИРОВАНИЕ СЛУЧАЙНЫХ ВЕЛИЧИН

EXPON FUNCTION RN2,C24

0,0/.1,.104/.2,.222/.3,.355/.4,.509/.5,.69/.6,.915/.7,1.2
.75,1.38/.8,1.6/.84,1.83/.88,2.12/.9,2.3/.92,2.52/.94,2.81
.95,2.99/.96,3.2/.97,3.5/.98,3.9/.99,4.6/.995,5.3/.998,6.2
.999,7/.9998,8

NORM FUNCTION RN3,C25

0,-5/.00003,-4/.00135,-3/.00621,-2.5/.02275,-2/.06681,-1.5
.11507,-1.2/.15899,-1/.21186,-.8/.27425,-.6/.34458,-.4
.42074,-.2/.5,0/.57926,.2/.65542,.4/.72575,.6/.78814,.8
.8413,1/.88493,1.2/.93319,1.5/.97725,2/.9979,2.5
.99865,3/.99997,4/1,5

DSB FUNCTION RN5,C2

0,50/.999999,130

GAUSS FVARIABLE 30+4*FN\$NORM

TAB1 TABLE IA,26,26,12

TAB2 TABLE FN\$DSB,110,50,3

TAB3 TABLE FN\$RAWN,57,7,12

TAB4 TABLE V\$GAUSS,18,2,12

GENERATE 67,FN\$EXPON

TABULATE TAB1

TABULATE TAB2

TABULATE TAB3

TABULATE TAB4

TERMINATE 1

START 500

END

Таблица 3.2.

Имя таблицы | Соотношения для С и В

Имя таблицы	Соотношения для С и В
TAB1	$C=4/\lambda*(D-2); B=C$
TAB3	$C=2r/d; B=m-r+C$
TAB4	$C=6*\lambda/(D-2); B=m-3*\lambda$

В соответствии с программой транзакты, создаваемые блоком GENERATE, проходят последовательно четыре блока TABULATE. При входе транзакта в блок TABULATE TAB1 осуществляется загрузка в таблицу 1 значения времени, прошедшего с момента входа в этот же блок предыдущего транзакта. Характеристики потока транзактов определяются операндами A и B блока GENERATE. Следует знать, что в рассматриваемой программе это - пуассоновский поток с интенсивностью $\lambda=0.015$. При входе транзакта в блок TABULATE TAB2 в таблицу TAB2 заносится значение дискретной псевдослучайной величины, разыгрываемой с помощью функции DSB. Аналогично в таблицы TAB3 и TAB4 заносятся значения псевдослучайных величин, определяемых с помощью RAWN и GAUSS соответственно.

В процессе реализации рассматриваемой GPSS программы через каждый блок TABULATE пройдут 500 транзактов и в каждую из таблиц TAB1-TAB4 будет занесено по 500 значений соответствующей псевдослучайной величины. Объем выборки в 500 испытаний задан в программе с помощью операнда A команды START. Выполнение GPSS программы прекратится, когда в блок TERMINATE войдет 500-й транзакт.

Порядок выполнения

1. Изучить GPSS-программы моделирования и методику расчета значений операндов B,C,D таблиц. Исходя из заданных в варианте работы распределений случайных величин, определить заданные функции, необходимые переменные, значения операндов D,C,B таблиц.
2. Написать и отладить программу.
3. По результатам моделирования построить графики эмпирических распределений моделируемых случайных величин.
4. Установить соответствие между теоретическими и эмпирическими функциями распределений.

Контрольные вопросы

1. Какая случайная величина называется базовой ?
2. Каким образом осуществляется имитация дискретных случайных величин ?
3. Как определяются в GPSS функции для разыгрывания значений дискретной и непрерывной случайной величины ?

4. Какие средства GPSS используются для имитации экспоненциально или нормально распределенных случайных величин ?
5. Пояснить назначение операндов A,B,C,D описания таблицы.
6. Какие данные о таблице выводятся на печать ?
7. Пояснить работу GPSS программы моделирования.

3.2.2. Моделирование ВС как СМО с одним обслуживающим прибором

В работе исследуется система обслуживания с прибором, имеющим один вход и один выход, которая является типовой моделью функционирования однопроцессорных ЭВМ, каналов связи, абонентских пунктов и т.д. Заявки на обслуживание в систему поступают в случайные или детерминированные моменты времени. Случайной или детерминированной величиной является и время обслуживания заявки.

Если прибор занят, то поступающие заявки обслуживаются в порядке очереди.

При подготовке к выполнению задания следует изучить принципы построения программных моделей СМО [1-4,6], описания следующих объектов и управляющих строк языка GPSS:

блоки - SEIZE A (ЗАНЯТЬ),
RELEASE A (ОСВОБОДИТЬ),
ADVANCE A,B (ЗАДЕРЖАТЬ),
QUEUE A,B (СТАТЬ В ОЧЕРЕДЬ),
DEPART A,B (ПОКИНУТЬ ОЧЕРЕДЬ),
команды - RESET (СБРОСИТЬ),
START A,B,C,D (НАЧАТЬ),
CLEAR (ОЧИСТИТЬ),

Содержание задания

Необходимо осуществить в соответствии с вариантом задания:

1. Программную имитацию следующих событий: создание заявки, постановку заявки в очередь, уход заявки из очереди, занятие прибора, задержку на обслуживание, освобождение прибора, уход заявки из системы.
2. Моделирование распределенной вероятности времени поступления и обслуживания заявок.

3. Разработать GPSS модель системы с учетом необходимости вывода на печать гистограмм распределения числа заявок, находящихся в системе (в очереди или на обслуживании), и времени нахождения заявок в системе.

4. Исследовать характеристики полученной модели в установившемся режиме при изменении параметров заданных функций распределения времени поступления или обслуживания заявок.

Примеры программ см. п.п. 2.1., 2.5

Контрольные вопросы

1. Привести примеры одноканальных СМО.
2. Какие средства GPSS используются для моделирования одноканальных СМО?
3. Какие характеристики системы можно получить, используя моделирование на GPSS?
4. Пояснить, как образуется поток Эрланга порядка k .
5. Пояснить работу GPSS программы моделирования системы с одним прибором.
6. Как определить время вхождения модели в установившийся режим?

3.2.3. Моделирование многопроцессорных ВС с перераспределением потока заявок

В работе исследуются модели многоканальных устройств (МКУ), которые выполняют параллельное обслуживание заявок. МКУ является группой однородных приборов. К ним относятся: микропроцессорная система с процессорами одного типа, совокупность дисплеев системы телеобработки, автоматическая телефонная станция и т.д.

При подготовке к выполнению задания необходимо изучить принципы построения математических и программных моделей МКУ [1-4,6,8], а также описания объектов языка GPSS:

блоков - ENTER A,B (ВОЙТИ),
LEAVE A,B (ВЫЙТИ),
TRANSFER A,B,C,D (ПЕРЕДАТЬ),

определений - имя STORAGE A (МНОГОКАНАЛЬНОЕ УСТРОЙСТВО).

Содержание задания

Необходимо осуществить в соответствии с вариантом задания:

1. Разработку GPSS модели системы обработки информации, состоящей из M микропроцессорных систем (МПС), включающих N однородных процессоров. Поток заданий на обработку является пуассоновским со значением среднего интервала K секунд. Время обработки задания процессором равно L секунд.

2. Исследование характеристик модели в установившемся режиме при изменении числа процессоров.

3. Определение оптимального числа процессоров в МПС, при котором 95% всех заданий на обработку выполняются.

Примеры программ см. в п.п. 2.3, 2.4, 2.6

Порядок выполнения

1. Составить GPSS программу моделирования системы обработки информации применительно к своему варианту исследования.

2. Осуществить один или несколько прогонов модели.

3. Определить оптимальное количество процессоров в МПС.

4. Оформить отчет в соответствии с установленными требованиями.

Контрольные вопросы

1. Привести примеры МКУ.

2. Какие средства GPSS используются для моделирования МКУ?

3. Какие режимы блока TRANSFER вы знаете?

4. Каким образом можно минимизировать число прогонов на ЭВМ для определения оптимального числа процессоров моделируемой системы?

5. Какие характеристики системы вы использовали при оптимизации?

6. Какие характеристики МКУ автоматически распечатываются после прогонов модели на ЭВМ?

3.2.4. Формирование потоков заявок в ВС

Работа посвящена моделированию источников заявок в вычислительных системах. Моделируются следующие характеристики заявок многомерного потока:

1. тип заявки - характеристика, указывающая важность заявки, т.е. ее приоритет;
2. длина - характеристика, определяющая размер заявки;
3. интенсивность потока заявок - характеристика, определяющая скорость образования заявок (количество появляющихся в ВС заявок за единицу времени);
4. место заявки - характеристика, указывающая адрес устройства, для которого она предназначена.

Часто тип заявки определяет особенности всех ее характеристик: интенсивность потока заявок данного типа, законы распределения длины заявки, адресов устройств для ее обработки.

В задачу моделирования источников многомерного потока заявок входит определение состава потока по типам заявок в соответствии с заданным законом распределения типов заявок, попадающих в поток; генерация заявок заданных типов в соответствии с законами распределения промежутка времени между поступлениями в систему заявок каждого типа; определение для каждой заявки ее длины и адреса устройства назначения, которые также случайны и в зависимости от типа заявки задаются соответствующим законом распределения вероятностей.

При подготовке к выполнению задания необходимо изучить описания объектов языка GPSS [6,8]:

операторов - ASSIGN A,B,C (НАЗНАЧИТЬ),
TRANSFER A,B,C,D (ПЕРЕДАТЬ),

особенности работы с параметрами транзактов.

Содержание задания

Выполнить моделирование источников сообщений N типов. Необходимые данные определяются вариантом задания. Получить поток из 100 заявок и оценить его характеристики.

Пример 3.2

Вариант задания представлен в табл.3.2.

Таблица 3.2

Распределения вероятностей

типов заяв	распр-е	адресов	заявок	длины (равн.)	промежут. врем.				
тип p(i)	1	2	3	4	5	a	b	(экспоненц.)	
1	0.26	0.05	0.62	0.0	0.1	0.23	10	120	lambda=0.015
2	0.28	0.12	0.57	0.19	0.03	0.09	33	76	lambda=0.025
3	0.46	0.19	0.14	0.5	0.17	0.0	45	340	lambda=0.05

Программа

```
* МОДЕЛИРОВАНИЕ МНОГОМЕРНОГО ПОТОКА ЗАЯВОК
* описание функций распределений и таблиц для сбора статистики
EXPON FUNCTION RN2,C24
0,0/.1,.104/.2,.222/.3,.355/.4,.509/.5,.69/.6,.915/.7,1.2
.75,1.38/.8,1.6/.84,1.83/.88,2.12/.9,2.3/.92,2.52/.94,2.81
.95,2.99/.96,3.2/.97,3.5/.98,3.9/.99,4.6/.995,5.3/.998,6.2
.999,7/.9998,8
FD1 FUNCTION RN3,D2 0,10/0.999999,120
FD2 FUNCTION RN3,D2 0,33/0.999999,76
FD3 FUNCTION RN3,D2 0,45/0.999999,340
TYPE FUNCTION RN5,D3 0.26,1/0.54,2/1,3
ADDR1 FUNCTION RN4,D5 0.05,1/0.67,2/0.67,3/0.77,4/1,5
ADDR2 FUNCTION RN4,D5 0.12,1/0.69,2/0.88,3/0.91,4/1,5
ADDR3 FUNCTION RN4,D5 0.19,1/0.33,2/0.83,3/1,4/1,5
TAB1 TABLE RT,1,21,12 статистика по интенсивности потока
TAB2 TABLE P1,1,1,5 статистика по типам заявок
TAB3 TABLE P2,10,15,22 статистика по длине заявок
TAB4 TABLE P3,1,1,6 статистика по адресации заявок
* формирование заданного потока заявок
GENERATE 67,FN$EXPON генерация заявок типа 1
ASSIGN 1,1 тип заявки в параметре 1
ASSIGN 2,FN$FD1 длина заявки в параметре 2
ASSIGN 3,FN$ADDR1 адрес заявки в параметре 3
```

TRANSFER .26,MODEL 26% заявок направл-ся в поток
TERMINATE остальные выв-ся из сист.
GENERATE 40, FN\$EXPON генерация заявок типа 2
ASSIGN 1,2
ASSIGN 2, FN\$FD2
ASSIGN 3, FN\$ADDR2
TRANSFER .28,MODEL
TERMINATE
GENERATE 20, FN\$EXPON генерация заявок типа 3
ASSIGN 1,3
ASSIGN 2, FN\$FD3
ASSIGN 3, FN\$ADDR3
TRANSFER .46,MODEL
TERMINATE

* точка модели, где сформировался заданный поток заявок
* и снимаются его характеристики

MODEL TABULATE TAB1 интенсивность потока
 TABULATE TAB2 распределение типов в потоке
 TABULATE TAB3 распределение длины заявок
 TABULATE TAB4 распределение адресации

* обработка заявок потока адресуемыми устройствами
* время обработки зависит от длины заявки

QUEUE P3
SEIZE P3
DEPART P3
ADVANCE P2, FN\$EXPON
RELEASE P3
TERMINATE 1
START 100 закончить работу
* после обработки 100 заявок

END

Контрольные вопросы

1. Изложить постановку задач моделирования источников заявок.
2. Чем характеризуются заявки ?
3. Является ли полученный поток простейшим ? И почему?

4. Объяснить различие между вычисленной и заданной вероятностями ?

5. Какие методы получения случайных величин с заданным законом распределения Вы использовали ? Почему ?

ЛИТЕРАТУРА

1. Бусленко М.П. Математическое моделирование производственных процессов на цифровых вычислительных машинах. - М.: Наука, 1964. - 364 с.

2. Мартин Дж. Системный анализ передачи данных: Пер. с англ. - М.: Мир, 1975.

3. Моделирование сложных дискретных систем на ЭВМ третьего поколения: (Опыт применения GPSS) / Голованов О.В., Дуванов С.Г., Смирнов В.Н. - М.: Энергия, 1978. - 160 с.

4. Шенон Р. Имитационное моделирование - искусство и наука.: Пер. с англ. - М.: Мир, 1978.

5. Майоров С.А., Новиков Г.И., Ариев Т.И. и др. Основы теории вычислительных систем. - М.: Машиностроение, 1978.

6. Шрайбер Т. Дж. Моделирование на GPSS / Пер. с англ. - М.: Машиностроение, 1980. - 592 с.

7. Прицкер А. Введение в имитационное моделирование и язык СЛАМ II: Пер. с англ. - М.: Мир, 1987.

8. Разработка САПР. В 10 кн. Кн. 9. Имитационное моделирование: Практ. пособие / В.М.Черненький; - М.: Высш. шк., 1990.